
Asynchronous dependency health checks library.
Features
	User defined dependencies health-checks with flexible settings, including configurable
health-check callback function timeout, startup delay and initial check state.
	Existence.Plug module providing customizable response for a http health-check endpoint.
	Support for multiple Existence instances and associated health-checks endpoints
(example use case: separate Kubernetes readiness and liveness http probes).
	Dependencies health-checks states are cached and accessed using a dedicated ETS table per
Existence instance, which means practically unlimited requests per second processing capacity
without putting unnecessary load on a health-checked dependency.

Installation
Add Existence library to your application mix.exs file:
# mix.exs
def deps do
  [
    {:existence, "~> 0.3.1"}
  ]
end
Usage
Define dependencies checks callback functions MFA's and start Existence child with your application
supervisor:
# lib/my_app/application.ex
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    health_checks = [
      check_postgres: %{mfa: {MyApp.Checks, :check_postgres, []}}
    ]

    children = [{Existence, checks: health_checks}]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
Declare your dependencies checks callback functions, in our example case PostgreSQL health-check:
# lib/my_app/checks.ex
defmodule MyApp.Checks do
  def check_postgres() do
    "SELECT 1;"
    |> MyApp.Repo.query()
    |> case do
      {:ok, %Postgrex.Result{num_rows: 1, rows: [[1]]}} -> :ok
      _ -> :error
    end
  end
end
Dependency health-check function check_postgres/0 will be spawned asynchronously by default every
30 seconds, checking if PostgreSQL instance associated with MyApp.Repo is healthy.
check_postgres/0 results will be cached in the ETS table used further to provide responses to user
requests. If /healthcheck endpoint requests are issued thousands of times per second, they do not
hit and overload PostgreSQL, instead cached results from ETS are returned.
Current overall health-check state can be examined with get_state/1:
iex> Existence.get_state()
:ok
Dependencies health-checks states can be examined with get_checks/1:
iex> Existence.get_checks()
[check_postgres: :ok]
Library provides Plug module generating responses to the /healthcheck endpoint requests.
Example Existence.Plug usage with Plug.Router.forward/2:
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  forward("/healthcheck", Existence.Plug)
end

Health-checks management and access module.
Module provides functions for accessing an overall health-check state and individual dependencies
checks results.
Module is also used to start an Existence process as a part of an application supervision tree.
Existence works by asynchronously spawning user defined dependencies checks functions.
Individual dependencies checks functions results are evaluated to establish an overall
health-check state.
Overall health-check state is healthy only when all user defined dependencies checks are healthy.
It is assumed that healthy state is represented by an :ok atom for both dependencies checks and
for the overall health-check.
Any other result in dependencies checks is associated with an unhealthy dependency check state.
Overall health-check unhealthy state is represented by an :error atom.
User defined dependencies checks functions are spawned as monitored isolated processes.
If user dependency check function raises, throws an error, timeouts or fails in any other way it
doesn't have a negative impact on other processes and it is gracefully handled by the library.
Current dependencies checks functions results and current overall health-check state are stored
in an ETS table.
Whenever user executes any of available state getters, request is made against ETS table which
has :read_concurrency set to true.
In practice it means that library can handle unlimited numbers of requests per second
without blocking any other processes, adding latency to the response or overloading dependency
with synchronous  requests.
Module provides following functions to access checks states:
	get_state/1 and get_state!/1 to get overall health-check state,
	get_checks/1 and get_checks!/1 to get dependencies checks states.

Functions with bangs are negligibly cheaper computationally because they don't check if ETS table
storing given Existence instance state exists and they will raise if such table doesn't exist.
Usage
After defining dependencies checks options, Existence can be started using
your application supervisor:
# lib/my_app/application.ex
def start(_type, _args) do
  health_checks = [
    # minimal dependency check configuration:
    check_1: %{
      mfa: {MyApp.Checks, :check_1, []}
    },
    # complete dependency check configuration:
    check_2: %{
      mfa: {MyApp.Checks, :check_2, []},
      initial_delay: 1_000,
      interval: 30_000,
      state: :ok,
      timeout: 1_000
    }
  ]

  children = [
    {Existence, checks: health_checks, state: :ok}
  ]

  opts = [strategy: :one_for_one, name: MyApp.Supervisor]
  Supervisor.start_link(children, opts)
end
When Existence is started it has assigned an initial overall health-check state, which
by default is equal to an :error atom, meaning an unhealthy state.
Initial overall health-check state can be changed with a :state key. In a code example above
initial overall health-check state is set to a healthy state with: state: :ok.
 Multiple Existence instances can be started by using common Elixir child identifiers
:id and :name, for example:
children = [
  {Existence, checks: readiness_checks,  id: ExistenceReadiness, name: ReadinessCheck},
  {Existence, checks: liveness_checks, name: {:local, LivenessCheck}}
]
Configuration
Existence options:
	:id - term used to identify the child specification internally. Please refer to the
Supervisor "Child specification" documentation section for details on child :id key.
Default: Existence.
	:name - name used to start Existence process locally. If defined as an atom()
:gen_statem.start_link/3 is used to start Existence process without registration.
If defined as a {:local, atom()} tuple, :gen_statem.start_link/4 is invoked and process is
registered locally with a given name.
Key value is used to select Existence instance when running any of the state getters,
for example: get_state(CustomName). Default: Existence.
	:checks - keyword list with user defined dependencies checks parameters, see description
below for details. Default: [].
	:state - initial overall Existence instance health-check state. Default: :error.
	:on_state_change - MFA tuple pointing at user function which will be synchronously applied
on the overall health-check state change.
User function should be of two arity. As a first argument it will receive current state as
:ok | :error atom. As a second argument function will receive static arg given in the MFA tuple.
Default: nil.

Dependencies checks are defined using a keyword list with configuration parameters defined
as a maps, see code example above.
Dependencies checks configuration options:
	:mfa - {module, function, arguments} tuple specifying user defined function to spawn when
executing given dependency check. Please refer to Kernel.spawn_monitor/3 documentation for
the MFA pattern explanation. Function will be spawned with arguments given in the :mfa key.
Required.
	:initial_delay - amount of time in milliseconds to wait before spawning a dependency check
for the first time. Can be used to wait for a dependency process to properly initialize before
executing dependency check function first time when application is started. Default: 100.
	:interval - time interval in milliseconds specifying how frequently given check should be
executed and dependency checked. Default: 30_000.
	:state - initial dependency check state when starting Existence. Default: :error.
	:timeout - after spawning dependency check function we will wait :timeout amount of
milliseconds for the dependency check function to complete.
If dependency check function will do not complete within a given timeout, dependency check
function process will be killed, and dependency check state will assume a :killed value.
Default: 5_000.

Dependencies checks
User defined dependencies checks functions must return an :ok atom within given :timeout
interval to acquire a healthy state.
Any other values returned by dependencies checks functions are considered as an unhealthy state.
Example health-checks callback functions for two popular dependencies, PostgreSQL and Redis:
#lib/checks.ex
def check_postgres() do
  "SELECT 1;"
  |> MyApp.Repo.query()
  |> case do
    {:ok, %Postgrex.Result{num_rows: 1, rows: [[1]]}} -> :ok
    _ -> :error
  end
end

def check_redis() do
  case MyApp.Redix.command(["PING"]) do
    {:ok, "PONG"} -> :ok
    _ -> :error
  end
end
Please notice that dependencies checks functions in the code example above are not wrapped in a
try/1 blocks.
Dependencies checks functions are spawned as monitored processes.
Whenever check function will raise, parent health-check process will be notified with an :info
:DOWN message and dependency check status will be assigned a tuple containing an exception and
a stack trace, for example:
# def check_1(), do: raise("CustomError")
iex> Existence.get_checks()
[
  check_1: {%RuntimeError{message: "CustomError"}, [ # ... stack trace ]}
]
iex> Existence.get_state()
:error

        get_checks(name \\ __MODULE__)

      


        Get dependencies checks states for name instance.



    


    
      
        get_checks!(name \\ __MODULE__)

      


        Same as get_checks/1 but raises if name instance doesn't exist.

        Get an overall health-check state for name instance.

          @spec get_checks(name :: atom()) :: [] | [{:key, :ok | any()}] | :undefined


      


Get dependencies checks states for name instance.
Function gets current dependencies checks states for an instance started with a given name,
by default Existence.
Dependencies checks functions results are returned as a keyword list.
If no checks were defined function will return an empty list.
Function returns :undefined if name instance doesn't exist.
Dependency check function result equal to an :ok atom means healthy state, any other term is
associated with an unhealthy state.
Example:
iex> Existence.get_checks()
[check_1: :ok, check_2: :ok]
iex> Existence.get_checks(NotExisting)
:undefined

          @spec get_checks!(name :: atom()) :: [] | [{:key, :ok | any()}]


      


Same as get_checks/1 but raises if name instance doesn't exist.
Function will raise with an ArgumentError exception if instance name doesn't exist.
Example:
iex> Existence.get_checks!()
[check_1: :ok, check_2: :ok]
iex> Existence.get_checks!(NotExisting)
** (ArgumentError) errors were found at the given arguments:

          @spec get_state(name :: atom()) :: :ok | :error | :undefined


      


Get an overall health-check state for name instance.
Function gets current overall health-check state for an instance started with a given name,
by default Existence.
Function returns an :ok when overall health-check state is healthy and an :error when state
is unhealthy.
Function returns :undefined if name instance doesn't exist.
Overall health-check state is healthy only when all dependencies health checks are healthy.
Example:
iex> Existence.get_state()
:ok
iex> Existence.get_state(NotExisting)
:undefined

          @spec get_state!(name :: atom()) :: :ok | :error


      


Same as get_state/1 but raises if name instance doesn't exist.
Function will raise with an ArgumentError exception if instance name doesn't exist.
Example:
iex> Existence.get_state!()
:ok
iex> Existence.get_state!(NotExisting)
** (ArgumentError) errors were found at the given arguments:

Plug responding with a health-check state.
Plug sends a text/plain response depending on the Existence overall health-check state
returned by an Existence.get_state/1 or Existence.get_state!/1 function.
Configuration
Plug is configured with a keyword list.
Plug response http status code and body are configurable with the following keys:
	:ok_status - response status code for the healthy state. Default: 200.
	:ok_body - response body for the healthy state. Default: "OK".
	:error_status - response status code for the unhealthy state. Default: 503.
	:error_body - response body for the unhealthy state. Default: "Service Unavailable".

Other configuration options:
	:raising? - if set to true plug will use Existence.get_state!/1 raising function to
get an overall health-check state.
If set to false, not raising Existence.get_state/1 function will be used.
Default: false.
	:name - instance name defined when starting Existence child with supervision tree.
Default: Existence.

Usage
Example module use with a Plug.Router.get/3 inside / route scope with a custom unhealthy
state response:
defmodule MyAppWeb.Router do
use MyAppWeb, :router

  #...
  scope "/", MyAppWeb do
    pipe_through(:browser)

    get(
      "/healthcheck",
      Existence.Plug,
      [error_status: 500, error_body: "Internal Server Error"],
      alias: false
    )
  end
end

Notice alias: false use to disable scoping on an external Existence.Plug in get/3 function.
Code example above will produce following http responses:
	healthy state:$> curl -i http://127.0.0.1:4000/healthcheck
HTTP/1.1 200 OK
...
OK

	unhealthy state:$> curl -i http://127.0.0.1:4000/healthcheck
HTTP/1.1 500 Internal Server Error
...
Internal Server Error

